
En primer lloc crearem una nova capa
anomenada buttons per sobre de l'herba,
on col·locarem els botons amb els qual
controlarem la línia de temps.

Per a aquests botons utilitzarem botons
ja creats que podem trobar seleccionant
Ventana > Bibliotecas comunes >
Botones. En un altre tutorial aprendrem
a modificar i crear els nostres propis botons, i n'explicarem en detall el
funcionament.

Arrosseguem a l'escenari els botons gel
Pause i gel Right que hi ha en la carpeta
classic buttons > Playback.

Ara aquests botons apareixeran en la
biblioteca del nostre document. Per tant, ja
podem tancar la biblioteca dels botons.

Seleccionem la instància de gel Pause en l'escenari i li donem el nom
d'instància pause_btn en l'inspector de Propiedades.

A la instància del botó gel Right, li donem el nom play_btn

Quan la línia de temps s'estigui reproduint volem que el botó per a
interrompre la pel·lícula sigui visible, i quan la pel·lícula estigui en pausa
necessitarem tenir visible el botó per a reprendre-la.

És a dir, farem que els botons es situïn al mateix lloc, i que es mostri l'un o
l'altre depenent de l'estat de la reproducció, com si fos un commutador.

Guia d'aprenentatge del Flash CS5

Tutorial 7 - Control de la línia de temps

En aquest tutorial afegirem dos botons per a reproduir i reprendre
la nostra animació.

Aprendrem a controlar tant la línia de temps principal d'una pel·lícula
com les línies de temps del clips imbricades en ella.

També aprendrem a controlar la reproducció d'animacions creades
amb programació.

Per fer aquest tutorial, obrim l'arxiu creat en el tutorial anterior
(tutorial6.fla) i el desem com a tutorial7.fla.

Seleccionem els botons amb la tecla Majúscules premuda per poder-los
seleccionar tots dos alhora. Els situem a prop de la cantonada inferior
dreta de l'escenari. Recordem que, en aquest cas, els límits de l'escenari
estan definits pel marc vermell.

Amb l'ajuda del panell Alinear, col·loquem tots dos botons al mateix lloc de
l'escenari, un a sobre de l'altre.

Obrim el panell Acciones (F9). Escriurem després de l'última línia que hem
programat en el tutorial anterior, en el fotograma 1 de la capa as.

Per diferenciar els dos tutorials i millorar la llegibilitat del codi, introduirem,
a més d'alguna línia en blanc, un comentari que separi les dues parts del
codi i que ens indiqui en quina zona de la programació som. Per exemple,
podem escriure el comentari següent:

//control buttons

Quan provem la pel·lícula, aquesta es reprodueix automàticament, per la
qual cosa el botó que volem que sigui visible en un primer moment és el
botó per a interrompre l'animació. Per tant, haurem de fer invisible el botó
play_btn. Per a això escriurem aquest codi:

play_btn.visible = false;

La propietat visible admet els valors true (visible) i false (invisible).

Un botó invisible no solament no es veurà, sinó que a més no estarà actiu.
Tanmateix, un botó amb un valor 0 d'alfa es podrà prémer encara que no
sigui visible. És per això que, quan volem desactivar un objecte a més de
fer-lo invisible, sigui preferible utilitzar la propietat visible en lloc de la
propietat alpha.

El pas següent és afegir detectors de l'esdeveniment de ratolí CLICK a tots
dos botons, cridant el botó play_btn a una funció que anomenarem
playMovie, i el botó pause_btn a una funció anomenada stopMovie:

pause_btn.addEventListener(MouseEvent.CLICK, stopMovie);

play_btn.addEventListener(MouseEvent.CLICK, playMovie);

La funció stopMovie pararà la pel·lícula. Quan la pel·lícula estigui aturada, el
botó per a reprendre la reproducció haurà de ser visible, mentre que ja no
caldrà mostrar el botó per a interrompre-la.

En el cas de la funció playMovie, en primer lloc reprendrà la reproducció de
la pel·lícula. Quan la pel·lícula es reprodueixi de nou, el botó per a
interrompre-la haurà de ser visible, però ja no caldrà que el botó per a
reproduir-la sigui visible.

Les funcions stop() i play() aturen i reprodueixen respectivament la línia
principal de temps.

Per tant, de moment les funcions stopMovie i playMovie quedaran de la
manera següent:

function stopMovie(e:MouseEvent):void

{

    stop();

    play_btn.visible = true;

    pause_btn.visible = false;

}

function playMovie(e:MouseEvent):void

{

    play();

    pause_btn.visible = true;

    play_btn.visible = false;

}

Si hi ha cap dubte amb aquesta part d'aquest codi, recomanem repassar el
tutorial anterior.

La línia play_btn.visible = false; que havíem col·locat abans de les
funcions (pas 3) s'executarà directament en provar la pel·lícula quan el cap
lector sigui en el primer fotograma. Tanmateix, la resta d'instruccions sobre
la propietat visible, en ser dins d'una funció, només s’executarà quan es
cridi la funció que les conté.

Seleccionem Control > Probar película (Ctrl+Intro).

Podem comprovar com s'alterna la visibilitat dels botons play_btn i
pause_btn quan els premem.

Veiem que prement pause_btn s’atura la línia de temps principal, és a dir,
l'avanç dels ocells. Tanmateix, les ales continuen en moviment, ja que la
funció stop() fa referència a la línia de temps principal, però no a les línies
de temps imbricades dins d'ella (com en aquest cas les ales).

D'altra banda, també continua l'avanç dels núvols per l'escenari, ja que el
seu moviment depèn de l'esdeveniment ENTER_FRAME, i aquest esdeveniment
és independent del desplaçament del cap lector en la línia de temps
principal.

Per a solucionar el problema del moviment de les ales, primer haurem de
saber com accedir-hi des de la programació en la línia de temps principal.

Per a accedir a una línia de temps imbricada dins d'una altra, en la
programació escriurem el nom de la instància principal, seguida d'un punt, i
seguida de l'element que està contingut en ella.

Per exemple, suposem que volem reproduir la línia de temps de la mà
dreta que hi ha dins del clip del braç dret que al seu torn pertany a un cos.
Per a això hauríem d'escriure el següent:

body.right_arm.right_hand.play();

D’aquesta manera la funció play() faria referència a aquesta mà. Per tant,
per a poder accedir a les ales de cada ocell, en primer lloc haurem de
donar un nom d'instància a cada ocell, i després a les ales.

Per a poder donar un nom d'instància als ocells, en primer lloc
desbloquegem les capes que els contenen.

Seleccionem cada ocell en l'escenari i els assignem els noms d'instància
bird1_mc, bird2_mc i bird3_mc, seguint la numeració dels noms de les
capes en les quals es troben.

Per a seleccionar els ocells 2 i 3 haurem de desplaçar el cap lector, ja que
no apareixen en l'escenari fins als fotogrames 24 i 150 respectivament.

Si fem clic dues vegades sobre qualsevol dels ocells de l'escenari,
entrarem en l'edició del clip bird original.

A l'interior, aquest clip de pel·lícula principal té dues instàncies del clip de
pel·lícula wing. Seleccionem l'ala esquerra i li donem el nom d'instància
wing1_mc. A l'ala dreta li donem el nom wing2_mc.

Tornem a l'escena principal prement Escena 1 en la barra d'edició, o bé
seleccionant-la en el desplegable.

Si premem dues vegades qualsevol dels altres ocells, veurem que ens
porta al mateix clip bird original, al qual ja hem donat els noms d'instància
wing1_mc i wing2_mc.

Provem d’aturar una ala d'un ocell. Per a això haurem d'afegir la línia de
codi següent dins de la funció stopMovie:

bird1_mc.wing1_mc.stop();

de tal manera que la funció stopMovie quedi com s’indica a continuació:

function stopMovie(e:MouseEvent):void

{

    stop(); 

    play_btn.visible = true;

    pause_btn.visible = false;

    bird1_mc.wing1_mc.stop();

}

Provem la pel·lícula i premem el botó pause_btn. L'ala esquerra del primer
ocell s’aturarà alhora que la línia de temps principal. Encara que
reprenguem la pel·lícula, aquesta ala no tornarà a reproduir-se, ja que no
hem escrit instruccions per a això.

Abans de continuar, provem de substituir en la programació bird1_mc per
bird3_mc, amb la qual cosa, la línia queda com s’indica a continuació:

bird3_mc.wing1_mc.stop();

Si tornem a provar la pel·lícula, i premem el botó pause_btn abans que
l'ocell 3 aparegui en escena, hi haurà l’error següent:

TypeError: Error #1009: No se puede acceder a una propiedad o a un 

método de una referencia a un objeto nulo.

    at tutorial7_fla::MainTimeline/detener()

Això vol dir que, en prémer el botó per a aturar la pel·lícula i executarse la
funció stopMovie, es fa referència a un objecte que no existeix en aquell
moment (bird3_mc en aquest cas), i per tant no pot executar la instrucció
d’aturar l'ala d'aquest ocell.

Malgrat que això no impedeix el funcionament correcte de la nostra
pel·lícula, per evitar aquest error comprovarem quins ocells són en
l'escenari en el moment de prémer els botons, i només executarem les
instruccions per aturar les ales d'aquests ocells.

Per a saber quins ocells hi ha en l'escenari en el moment de prémer un botó
cal saber en primer lloc en quin fotograma som en aquest moment.

Per a saber en quin fotograma és el cap lector en un moment donat,
utilitzarem la propietat currentFrame.

Sabem que l'ocell 1 hi és des del primer fotograma i que desapareix en el
fotograma 500. Per tant, si la propietat currentFrame torna un nombre més
petit que 501, l'ocell 1 estarà en escena.

L'ocell 2 apareix en el fotograma 24 i desapareix en en el 524, així que, si
el cap lector és entre aquests dos valors, també podrem assegurar que
l'ocell 2 és en l'escenari. De la mateixa manera, l'ocell 3 és present entre els
fotogrames 150 i 350.

Per considerar que un ocell està en escena no és necessari que sigui dins
del marc visible de l'escenari, sinó que el que es té en compte és que
estigui present en la línia de temps.

Si tenim en compte que < significa més petit que, > significa més gran que,
i && equival a l'AND lògic, la funció stopMovie quedarà com segueix:

function stopMovie(e:MouseEvent):void

{

    stop(); 

    play_btn.visible = true;

    pause_btn.visible = false;

    

    if (currentFrame < 501) 

    {

        bird1_mc.wing1_mc.stop();

        bird1_mc.wing2_mc.stop();

    }

    

    if (currentFrame > 23 && currentFrame < 525) 

    {

        bird2_mc.wing1_mc.stop();

        bird2_mc.wing2_mc.stop();

    }

    

    if (currentFrame > 149 && currentFrame < 351) 

    {

        bird3_mc.wing1_mc.stop();

        bird3_mc.wing2_mc.stop();

    }

}

Explicarem un dels condicionals per a comprendre millor el codi. Per
exemple, amb la condició

if (currentFrame > 23 && currentFrame < 525)

comprovem si el fotograma actual és més gran que el fotograma 23 i, a
més, si és més petit que el fotograma 525.

Si es compleixen tots dos requisits, llavors la condició completa es
compleix, i s'executaran les instruccions que hi hagi entre les claus d'aquest
if. Si alguna de les parts de la condició no es compleix, no s'executaran les
instruccions corresponents.

Segons en quin fotograma siguem, pot ser que es compleixin les
condicions que hem escrit per a tots els ocells (per exemple, el fotograma
200), només per a algun (per exemple, el fotograma 15), o per a cap dels
tres (per exemple, el fotograma 580).

Si provem la pel·lícula, veurem que ara es paren correctamente les ales
dels tres ocells, i no se'ns informa de cap error. Ara ens falta tornar a
activar el moviment de les ales quan tornem a reproduir la pel·lícula.

Per tornar a activar el moviment de les ales, copiem en la funció playMovie
les sentències condicionals que hem utilitzat en la funció stopMovie.
Després, simplement substituïm el mètode stop() pel mètode play(), per
exemple:

if (currentFrame < 501) 

{

    bird1_mc.wing1_mc.play();

    bird1_mc.wing2_mc.play();

}

Tornem a provar la pel·lícula. Ara amb els botons play_mc i pause_mc podem
aturar i reprendre tant l'avanç dels ocells per l'escenari, com el moviment
dels ales.

A més de parar amb stop() i reprendre amb play(), hi ha altres mètodes
interessants a l'hora de controlar una línia de temps com, per exemple:

nextFrame() i prevFrame(), per a avançar i retrocedir un fotograma

gotoAndPlay() i gotoAndStop(), per a anar a un fotograma concret, i una
vegada allà continuar o aturar-nos, i escriure dins dels paréntesis el
número de fotograma al qual volem anar o el seu nom (etiqueta).

Fins aquí hem vist com podem controlar el cap lector tant de la línia de
temps principal com de les línies de temps imbricades.

Ara ens queda aturar i reprendre el moviment dels núvols, ja que aquest
moviment és independent del desplaçament del cap lector en la línia de
temps principal.

El moviment de cada núvol l'havíem creat en el tutorial anterior amb el codi
següent:

cloud1_mc.addEventListener(Event.ENTER_FRAME, wind);

cloud2_mc.addEventListener(Event.ENTER_FRAME, wind);

Perquè els núvols s’aturin, n’hi haurà prou d'eliminar els detectors de
l'esdeveniment ENTER_FRAME. Per a eliminar-los n'hi ha prou de copiar el
mateix codi que es va usar per a crear-los, i substituir addEventListener per
removeEventListener.

cloud1_mc.removeEventListener(Event.ENTER_FRAME, wind);

cloud2_mc.removeEventListener(Event.ENTER_FRAME, wind);

Per tant, afegirem les dues sentències amb removeEventListener a la funció
stopMovie abans de l'última clau que marca el final de la funció. Perquè el
moviment es reprengui després, tornarem a afegir en la funció playMovie
les dues sentències amb addEventListener.

Si, a més d'aturar el moviment, volguéssim que els núvols deixessin de ser
arrossegables, hauríem de seguir el mateix procediment, és a dir, copiar la
mateixa sentència que utilitzem per a agregar el listener i substituir add per
remove. En aquest cas permetrem que els núvols puguin arrossegar-se
encara que la línia de temps estigui aturada, així que no eliminarem aquests
listeners.

El mètode removeEventListener pot ser molt útil en diverses circumstàncies,
per exemple, per a eliminar la interactivitat d'un botó o un clip sense
necessitat d'amagar-lo ni eliminar-lo de l’escenari.

Seleccionem Control > Probar película. Ara podem aturar i reprendre
totes les animacions (ocells, ales i núvols).

Per a complementar els conceptes desenvolupats en aquest tutorial, es
recomana fer les activitats següents:

Afegeix botons per a avançar i retrocedir un fotograma.1.

Afegeix un botó que vagi al principi de la pel·lícula.2.

Fes que els núvols es puguin arrossegar només quan la pel·lícula
estigui aturada.

3.

